
Introduction 
ProbCalc was originally developed as a one off, Google Sheets-based election predictor 
during November 2025. Since then, it has expanded to cover sports games, the complexities 
of constituency-based First Past the Post, and added a more complex winner identification 
system. 
 
ProbCalc uses weighted polls to simulate election outcomes, and this is what the guidebook 
will seek to teach you. 

I. Gamma distribution 
In order to model uncertainty in election outcomes, ProbCalc treats polling data not just as 
fixed values, but as a range of weighted inputs used to generate different election outcomes. 
To do this, it uses a gamma-based curve, specifically the inverse gamma form. We’ll define 
three main parameters: α for shape, β for scale, and p for position. 

 
The exact details of this function do not matter as much as the parameters we can change. I 
will only define these important parameters. 
 
 

●​ The bigger an α is in an inverse gamma function, the more 
concentrated the distribution becomes. If our α=4, β=1 and 
p=0.5, the output of inputting these parameters into the inv. 
gamma function would equal ~3.67, as shown in the graph. 
If the α is 100, then the curve would be much steeper, see 
second graph. The result equals ~99.67. 
Note: α CANNOT be smaller than 0, since the gamma 
function is undefined for non-positive shape values 
 

 
●​ β defines the scale of our curve. It basically zooms in or out 

on the graph and the values. Let's go back to my first 
example and shift β up to 11. Since our β is at 11, we've 
zoomed out to the point where the original x axis of the 
graph is just 1/11th of the new one, while the shape remains 
unchanged. Shocker.  
Note: β CANNOT be smaller than 0. 

 
 

●​ p or position defines how far along a certain curve the answer or output is. For 
instance, if p=0.2 in example 3, the output would equal 25.26, as shown in the graph. 
Note: p HAS to be within this range: 0<p<1 

 
Now that we understand how the gamma distribution behaves, we will move on to how we 
calculate our α in BaseCalc. 

II: Simple ProbCalc 

II.I: BaseCalc 
This section will go over how I get α from polling data, and each section will include practical 
exercises at the end to test your understanding. 
 
To produce a reliable polling aggregate, we will need to take time of release and sample size 
into account. Recent polls are given more influence in the average than polls released 
earlier. Similarly, polls with larger sample sizes contribute more to the final result.  
This keeps the average reflects the modern day as opposed to outdated polling data.  

II.I.i: Data collection and formatting 
All of my works as of April 2026 have used Wikipedia as the main and only source of polling 
data. Opinion polling sections/pages are a great source of the individual polls with very little 
clutter, so I just copy-paste it straight from Wikipedia to Google Sheets.  
The important thing is to only select up to a certain pivotal point (e.g. start of the year, major 
changes in candidates/parties). I cut off the polling at the date when the Progressive 
Bulgaria party was founded, as including earlier polls introduces inconsistencies in the 
dataset. 

 
When you copy-paste it from Wikipedia, the table should look something akin to this. Even 
with the clarity of the dataset, some formatting is still required. 

●​ Remove Wikipedia's event cells. This makes the dataset easier to follow, and 
alleviates part of the workload later down the line. 

●​ Make sure all of your inputted data registers as numbers. For dates, use either the 
date of publication (used in the example above) or do an average. Averaging the 
date hurts polls that collect data for longer. 

●​ Be aware of region-based differences in how Google Sheets formats numbers. For 
the Europeans, you'll need to change a lot more than the Americans will. This means 
that a period (.) in American Google Sheets separates wholes and decimals, while 
Europeans use a comma(,). Make sure to adjust the data in accordance  

II.I.ii: Calculations 

 
The next step is to copy and paste the whole sheet, again. This time you'll copy your own 
formatted one to save you the hassle and perform calculations without changing the original 
data. 
Under the “Fieldwork Date” column, change the formula to 
=[ELECTIONDAY]-[ORIGINALFIELDWORKDATE]. Enter the election day as a date 
(convert it if necessary), then format it as a regular number. In practice your formula should 
look something like this: =46131-B2. Copy and paste this to every “Fieldwork date” cell in 
the new table. 
 
For the individual poll results, I use this formula: 
=([POLLRESULT]*[SAMPLESIZE])/([DAYSTILLELECTION]*100). Multiplying by 
sample size gives greater weight to larger polls. I subsequently divide the output by the 
distance from election day to publication to keep the data relevant to the modern day. The 
*100 keeps α lower, which increases variation in the simulation. 
 
 Anchor the [SAMPLESIZE] column with a $ sign so that it stays in place when you copy 
and paste the formula. Do the same for the [DAYSTILLELECTION] parameter. The end 
result applied practically should end up looking like this:  
=(D3*$C3)/($B27*100).  
When spreading the formula to cover the whole poll sheet, you can paste the formula only 
and keep the format via Paste Special > Formula Only, which is more aesthetically pleasing 
in my opinion. 

 
You will then sum the values together through the SUM function, e.g. 
=SUM([RANGEBORDER1]:[RANGEBORDER2]). Practically, it looks like this: 
=SUM(D27:D47). The function adds everything that you throw at it together. Copy paste it 
column to column without anchoring.  
This is your α parameter.  For the percentage, take your first cell, in my case GERB-SDS, 
and divide it by the sum of all alphas, so 
=[CELL]/SUM([RANGEBORDER1]:[RANGEBORDER2]). Make sure to anchor your 
[RANGEBORDER1] and [RANGEBORDER2], as without it, the cells would shift upon 

copy-pasting. It looks like this in practice: =D52/SUM($D52:$O52). Apply this to each cell, 
and you have not only your alpha, but your poll average as well. 
 

This is just how it looks put into a pie chart. 

II.I.iii: Exercise 
Take the data points from 2024 Pennsylvania Senate polling, and use the September 10th 
Presidential Debate as your cutoff date. Reformat the data and calculate the poll average.  
 
Your average should come out to roughly 48.7% Casey – 45.5% McCormick 
This, while not reflecting the real results, shows us that polls and their averages are not 
always correct. 

II.II: ProbCalc 

II.II.i: First simulation 
Create a new sheet to simulate the elections. Start by copying and pasting the individual 
party names into a vertical format, and fetch the alphas with this formula: 
=[SHEETNAME!CELL]. In practice, this would be something like =BaseCalc!D51.  

Next, apply the inverse gamma distribution. In Google Sheets this is written as: 
=GAMMA.INV([p];[α];[β]). Recall that p defines the position along the distribution, 
and we ought to make that random using RAND().Your α variable should be the BaseCalc 
result we carried over, whereas β should remain at either 1 or 0.1. Your ProbCalc formula 

should look like this: 
=GAMMA.INV(RAND(); 
$C$1; 1). Remember to 
anchor both your column 
and row identifiers in the α, 
and manually adjust your 
first simulation’s row 
reference to correspond 
with the actual data by 
party. 
 
The percentage rows can 
then be calculated by 
following the steps in the 
BaseCalc section. 

II.II.ii: Identifying the winner and probability 
The most crucial thing on the list right now is actually counting who won.  
For this, we will use the INDEX, MATCH, and MAX functions. The formula should be placed in 
the beginning cell of the simulation:  
=INDEX([PARTYNAMEBEGIN]:[PARTYNAMEEND];MATCH(MAX([GAMMABORDER1]:[G
AMMABORDER2]); [GAMMABORDER1]:[GAMMABORDER2]; 0)) 

This may look complex at first, but it is straightforward.  
●​ MAX finds the highest number in the output range, but does not give the actual cell 

number.  
●​ That’s what MATCH is for. It finds the same exact value in the range and returns its 

position.  
●​ This is used by INDEX to provide the name of the winning party/candidate.  

In practice, it looks like this: =INDEX($A$1:$A$12; MATCH(MAX(D14:D25); 
D14:D25; 0)). Apply this formula to identify the winner of a simulated result. Now, copy 
and paste the simulations from your simulation #1 down to row 100. 
The final step of creating a basic ProbCalc is to count the number of victories. For 
readability, I would recommend taking the party names and colours (again) and pasting them 
to the side. We will use the COUNTIF function like so: 
=COUNTIF([COLUMNSTART]:[COLUMNEND]; [PARTYNAMECELL]). The first part of the 
bracket defines the range while the second part looks in that range to find and count any 
matches with the defined text. In application, it looks like this =COUNTIF(D:D; F2). 
Copy and paste until you have all of your parties covered, and if desired, add a percentage 
calculator. 

 
II.II.iii: Exercise 
Open your Pennsylvania US Senate 2024 sheet and find a scenario where David 
McCormick wins with more than 60% of the vote. Do this by adjusting either the scale factor 
or shrinking the α. 

III. Complex ProbCalc 
In this section, I will lay out three more complex formulas and go over what they actually do. 
This part is slightly more theoretical, as I am assuming that you are already comfortable with 
this kind of sheet setup.  

III.I: Structured majority identification 
We will extend our previous winner-identification formula to also include the type of victory 
(plurality, majority, or supermajority).  
 
=INDEX([PARTYNAMEBEGIN]:[PARTYNAMEEND];MATCH(MAX([%RANGEBORDER1]:[
%RANGEBORDER2]); [%RANGEBORDER1]:[%RANGEBORDER2]; 0))  
 & " — " & 
IF(MAX([%RANGEBORDER1]:[%RANGEBORDER2])>=0,66;"Supermajority";IF(M
AX([%RANGEBORDER1]:[%RANGEBORDER2])>=0,5;"Majority";"Plurality")) 
I’ll admit, this does look very scary at first, but it only consists of two major parts:  

●​ In the first part of our formula, we identify the winner. 
●​ In the second, we determine by how much they win. 
●​ All of this is joined by an em dash or hyphen. 

Our [%RANGEBORDER1]:[%RANGEBORDER2] parameter contains the percentage or seat 
values.  
IF(MAX([%RANGEBORDER1]:[%RANGEBORDER2])>=0,66;"Supermajority"; 
checks whether the largest percentage value is exceeds 66%, and if so, it types out [Winner] 
– Supermajority.  
We then get to our ELSE function, which we will split to another IF tree that defines if the 
winner has more than 50%. If it finds the winner has over 50%, it writes out [Winner] – 
Majority, and if they don’t, then it writes out [Winner] – Plurality.  
If you want to do this in a parliamentary simulation, adjust the numbers in the IF tree to 
match the seats required. In practice, it looks like this: 
=INDEX($A$1:$A$12; MATCH(MAX(D14:D25); D14:D25; 0)) & "-" & 
IF(MAX(D14:D25)>=0,66;"Supermajority";IF(MAX(D14:D25)>=0,5;"Majori
ty";"Plurality")) 
In order to count the victories by type and winner, update your COUNTIF to match the output, 
e.g. instead of =COUNTIF(D:D; F2), use the actual output =COUNTIF(D:D; F2 & 
“Plurality”). This takes very little manual work compared to some other methods, it is 
recommended to utilise this method. It is also advised to organize these into a table for 
readability.  

III.I.i: Exercise 
Returning back to the Pennsylvania 2024 US Senate ProbCalc, build a similar system, but 
for widely-used predictor margins/probabilities: Safe, Likely, Lean and Tilt. 

III.II: Split vote count and extrapolation 
Now, we will tackle how to actually count results by smaller groups, such as demographics 
or administrative divisions. Firstly, get your by-group data, and format it so that the parties 
align and ideally, that there’s an equal number in the previous election result and the election 
you’re trying to simulate. 
I’ll make a mock spreadsheet like this: 

 
 

For the sake of simplicity, there’s 
only two parties and three main 
groups. We will only calculate the 
percentage. You can, however, do 
this via actual vote counts, though 
these are generally less accurate 
than the percentages themselves. 
What I’m going to do now is 
calculate the relative margin of a 
group compared to the topline 
margin by subtracting the group 
result from the topline. 
=[TOPLINE]-[GROUPRESULT] 
With this piece of data, we can just 
add this adjustment to our 
BaseCalc in order to get our new 

results. 
=[BASECALC]+[COMPARISON]. This formula is also used when counting parliamentary 
seats, especially those without local polling. 

III.II.i: Getting a feasible vote count 
Take the sum of your alpha’s and multiply it by the division of the vote count last election and 
the alpha. =[BASECALC]*([VOTECOUNT]/[BASECALC]). Use this after your gamma 
simulation. 

III.II.ii: Exercise 
Calculate the expected 2024 Pennsylvania US Senate margin of Bob Casey with 
college-educated voters. Use 2022 data for the past election data. 

​
III.III: Counting the votes 
In this section, we will go over a simple formula that allows you to count votes in each and 
every simulation. 
=SUM(FILTER([RANGEBORDER1]:[RANGEBORDER2]; 
MOD(SEQUENCE(ROWS([RANGEBORDER1]:[RANGEBORDER2]));[NUMBEROFPARTIES
+1IFBLANKSPACE])=1)) 
Let’s go over all of this step by step.  

●​ The ROWS function counts how much rows there is in a given range, 
●​ The SEQUENCE function “lists” the number of rows, e.g. 1,2,3… 
●​ The MOD function, short for Modular Arithmetics, points out whether or not row 

numbers are an exact multiple of the given number 
[NUMBEROFPARTIES+1IFBLANKSPACE]. 

●​ The FILTER function with the =1 parameter at the end filters out only the cells for 
which the MOD function is “True” 

●​ And finally, the SUM function counts all of these cell values together. 
●​ Combining these 6 functions together gives us a master formula for counting a 

party’s votes in a vertical simulation system, alternatively, you can simply reformat 
your simulations/calculations to be in a horizontal format, in which you can simply 
use the SUM function to count the votes. 

Here’s what this formula looks like in practice: 
=SUM(FILTER($H4:$H$1118; MOD(SEQUENCE(ROWS($H4:$H$1118));7)=1)) 
Make sure to anchor some points, like columns and [RANGEBORDER2] cell numbers for 
easier copy-pasting. 

III.III.i: Exercise 
Run and count the votes by education level and then at-large in the 2024 Pennsylvania US 
Senate Election. 

